



プログラミング演習2 ファイル入出力

中村, 小松, 小林, 橋本



- クッキークリッカーって知ってる？
 - 「クッキークリッカー」で検索
 - ただひたすらクリックしてクッキーを焼いていくゲーム
 - クリックして放置するだけのインフレゲーム

**クリックするの面倒くさい！
コンピュータに任せたい！**



- ロボットクラスだ！

- ひたすらマウスカーソルの位置をクリックするプログラムを書いたらどうだろうか？
- 書いてみよう！（演習）

- 左ボタン

- InputEvent.BUTTON1_MASK

```
import java.awt.Robot;
import java.awt.event.*;
Robot myRobot;
void setup(){
    size( 50, 50 );
    try {
        myRobot = new Robot();
    } catch( Exception e ){
    }
}

void draw(){
    // ここでクリック！
}
```



- データはファイルから読み込み，プログラム自体は変更しない！
 - 設定や敵の強さなど

データをファイルから
読み込むにはどうするか？



・ ファイルから読み込むとは？

C:¥ どこかのフォルダ **¥** どこかのフォルダ **¥** なんかのファイル

上
か
ら
順
に
読
み
込
む

こんにちは みさなん おんげき ですか？

わたしは げんき です。

この ぶんしょう はいりぎす の ケブンツリジ
だ が いく の けゅきんう の けっか

に ん ん げ は もじ を に しんき する とき その
さ しい よ と さいご の もさじ え あいて っれば
じ ばん ゆん は めく ち ちゃ や でも ちん や と
よ める という けゅきんう に もづい と て わざ と
もじ の じん ば ゆん を い か れ え て あ ま り す。
ど で う す？ ちん や と よ や ち め う で し ょ？
ちん や と よ た め ら は ゆ し く を よ し ろ く

ファイルから読み込む



```
String [] lines = loadStrings("ファイル名");
```

- ファイルの中身を1行毎にStringの配列に格納
 - 1行目の値は lines[0] に, 2行目の値は lines[1] に入っている
- **int(文字列)** で, 文字列を整数に変換

```
String [] lines = loadStrings("list.txt");  
int name = lines[0];  
int cost = int( lines[1] );
```

駅名 + 値段を表示する

明治大学総合数理学部
先端メディアサイエンス学科
中村研究室



駅名 + 値段を表示するプログラムを作成する。
なお、駅名と値段の情報は list.txt から読み込むものとする

ticket_bending_machine	
Ochanomizu	130
Iida-bash	150
Shinjuku	190
Nakano	210
Kichijoji	280

ファイルを読み込む



list.txt (駅名, 値段が各行に)

Ochanomizu

130

Iida-bash

150

Shinjuku

190

Nakano

210

Kichijoji

280

Ochanomizu	130
Iida-bash	150
Shinjuku	190
Nakano	210
Kichijoji	280

```
String [] names = new String [5];  
int [] prices = new int [5];
```

```
void setup() {  
    size( 400, 400 );  
    String [] lines = loadStrings( "list.txt" );  
    textSize( 30 );  
    names[0] = lines[0];  
    prices[0] = int(lines[1]);  
    names[1] = lines[2];  
    prices[1] = int(lines[3]);  
    names[2] = lines[4];  
    prices[2] = int(lines[5]);  
    names[3] = lines[6];  
    prices[3] = int(lines[7]);  
    names[4] = lines[8];  
    prices[4] = int(lines[9]);  
}
```

```
void draw() {  
    background(255);  
    fill(0);  
    for ( int i=0; i<5; i++ ) {  
        text( names[i], 50, i*50+50 );  
        text( prices[i], 300, i*50+50 );  
    }  
}
```



ファイルを読み込む



```
String [] names = new String [5];  
int [] prices = new int [5];
```

```
void setup() {  
    size( 400, 400 );  
    String [] lines = loadStrings( "list.txt" );  
    textSize( 30 );  
    for( int i=0; i<5; i++ ){  
        names[i] = lines[i*2];  
        prices[i] = int(lines[i*2+1]);  
    }  
}
```

```
void draw() {  
    background(255);  
    fill(0);  
    for ( int i=0; i<5; i++ ) {  
        text( names[i], 50, i*50+50 );  
        text( prices[i], 300, i*50+50 );  
    }  
}
```

ファイルを読み込む



- ファイルを修正して1行を1つの駅名と値段に

```
Ochanomizu  
130  
Iida-bash  
150  
Shinjuku  
190  
Nakano  
210  
Kichijoji  
280
```

list.txt



```
Ochanomizu,130  
Iida-bash,150  
Shinjuku,190  
Nakano,210  
Kichijoji,280
```

list.csv

カンマ区切りで書く

ファイルを読み込む



```
String [] names = new String [5];  
int [] prices = new int [5];  
  
void setup() {  
    size( 400, 400 );  
    String [] lines = loadStrings( "list.csv" );  
    textSize( 30 );  
    for( int i=0; i<5; i++ ){  
        String [] data = split( lines[i], ',' );  
        names[i] = data[0];  
        prices[i] = int( data[1] );  
    }  
}
```

**String [] 文字列配列 =
split(文字列, '区切り文字');**

```
String [] names;  
int [] prices;
```

```
void setup() {  
  size( 400, 400 );  
  String [] lines = loadStrings( "list.csv" );  
  names = new String [lines.length];  
  prices = new int [lines.length];  
  textSize( 30 );  
  for( int i=0; i<lines.length; i++ ){  
    String [] data = split( lines[i], ',' );  
    names[i] = data[0];  
    prices[i] = int( data[1] );  
  }  
}
```

String [] 文字列の配列
= split(文字列, '区切り文字');

**配列.length で
配列のサイズ**

```
void draw() {  
  background(255);  
  fill(0);  
  for ( int i=0; i<names.length; i++ ){  
    text( names[i], 50, i*50+50 );  
    text( prices[i], 300, i*50+50 );  
  }  
}
```



- list.csv ファイルを書き換えて，値を追加したり，色々な値を読み込んでみましょう

```
Ochanomizu,130  
Iida-bash,150  
Shinjuku,190  
Nakano,210  
Kichijoji,280  
Ogikubo,290  
Mitaka,380  
Kokubunji,540  
Nishi-kokubunji,540  
Tachikawa,620
```

ticket_bending_machine	
Ochanomizu	130
Iida-bash	150
Shinjuku	190
Nakano	210
Kichijoji	280
Ogikubo	290
Mitaka	380
Kokubunii	540

セーブ&ロード



- 前回起動していたときの状態を引き継ぐには、何らかの形でデータをセーブし、ロードしてデータを読み込む！
- さて、どうやって保存して読み込む？

おきのどくですが
ぼうけんのしょ 1 ばんは
きえてしまいました。



- これまで
 - 変数に値を保存（代入）し，変数を使うことで値を取り出し表示したりしていた
 - Ballのx, y座標と，スピード
 - ゲームのscore
 - 電光掲示板のライトのON/OFF情報

アプリケーションを再起動したら消えてしまう



ファイルに記録して再起動後にも使う！

ファイルに保存する



- 現在の状態をファイルに保存
 - $x = 10$
 - $y = 20$
- ファイル（data.txt）に例えば下記のように保存

10

20

- 1行目は x , 2行目は y の値として読み込む



```
saveStrings("ファイル名", String型の配列);
```

- String型の配列の内容を，1行毎にファイルに書き込む
 - 例えば，Stringの配列をlinesとしたときに，lines[0]は1行目に，lines[1]は2行目に保存される



電光掲示板プログラム

明治大学総合数理学部
先端メディアサイエンス学科
中村研究室



横に10個，縦に1個並んだ電光掲示板（画面サイズは300x100とする）について，丸をクリックする度に赤色，白色と塗りつぶしの入れ替わるようにせよ．また，ファイルから赤白の状態を読み込むようにせよ

- メモ帳で，1行に0または1だけを書いた10行のファイルを作成する（ファイル名はstatus.txt）
 - PDEと同じフォルダに保存する
- status.txt をプログラムで読み込み，1行目をlights[0]に，2行目をlights[1]にと値を順に割り当てる
- lightsの値に応じて描画





```
int [] lights = new int [10];
void setup() {
  size( 300, 100 );
  int i=0;
  while( i<10 ){
    lights[i] = 0;
    i++;
  }
}

void draw() {
  background( 255 );
  int i=0;
  while( i<10 ){
    if ( lights[i] == 1 ) {
      fill( 255, 0, 0 );
    }
    else {
      fill( 255 );
    }
    ellipse( 15+30*i, 50, 30, 30 );
    i++;
  }
}
```

```
void mousePressed(){
  int i=0;
  while( i<10 ){
    if( dist(15+30*i,50,mouseX,mouseY)<=15 ){
      if( lights[i] == 1 ){
        lights[i] = 0;
      } else {
        lights[i] = 1;
      }
    }
    i++;
  }
}
```


電光掲示板プログラム

明治大学総合数理学部
先端メディアサイエンス学科
中村研究室



- ファイルから読み込む

```
String [] lines = loadStrings("status.txt");
```



配列	値
lines[0]	"0"
lines[1]	"1"
lines[2]	"0"
lines[3]	"0"
lines[4]	"1"
lines[5]	"0"
lines[6]	"0"
lines[7]	"1"
lines[8]	"0"
lines[9]	"1"



setup()のみを変更してファイルから読み込む！

```
int [] lights = new int [10];
void setup() {
  size( 300, 100 );
  String [] lines = loadStrings( "status.txt" );
  lights[0] = int( lines[0] );
  lights[1] = int( lines[1] );
  lights[2] = int( lines[2] );
  lights[3] = int( lines[3] );
  lights[4] = int( lines[4] );
  lights[5] = int( lines[5] );
  lights[6] = int( lines[6] );
  lights[7] = int( lines[7] );
  lights[8] = int( lines[8] );
  lights[9] = int( lines[9] );
}
```




```
int [] lights = new int [10];
void setup() {
  size( 300, 100 );
  String [] lines = loadStrings( "status.txt" );
  int i=0;
  while( i<10 ){
    lights[i] = int( lines[i] );
    i++;
  }
}
```

こんなエラーが出たら



- status.txt がPDEと同じフォルダに入っていないというエラー



電光掲示板プログラム

明治大学総合数理学部
先端メディアサイエンス学科
中村研究室



```
void mousePressed(){
    int i=0;
    while( i<10 ){
        if( dist(15+30*i,50,mouseX,mouseY)<=15 ){
            if( lights[i] == 1 ){
                lights[i] = 0;
            } else {
                lights[i] = 1;
            }
        }
        i++;
    }


    i=0;
    String [] lines = new String [10];
    while( i<10 ){
        lines[i] = str( lights[i] );
        i++;
    }
    saveStrings( "status.txt", lines );
}
```

演習 (lights_3x3)



(Q4) Q3の電光掲示板のプログラムを改良して縦に3個，横に3個となるように変更し，左下のように記述したファイルから状態を読み込み下図のように出力せよ．また，値を変えて動作を確認せよ

1, 0, 1
0, 1, 0
0, 0, 1



演習 (lights_3x3)



ヒント


- 3x3の配列を作る
- 1行ずつ取得し，そのそれぞれの値を配列に代入していく
- 1行の値を読み込む時は，カンマ区切りであることを考慮
 - カンマごとに区切る場合は `split` を使う

おまけ: 成績グラフ作成



- <http://snakamura.org/teach/fms/scores.txt> という180人分の成績データ（架空）を読み込んで，点数毎の頻度分布を作ってみましょう

- 0-10までの数を数える配列を作成する
- ファイルから読み込んだ値に応じて，配列のそれぞれの数を1ずつ増やしていく





```
String [] lines = loadStrings("URL");
```

```
int [] scores = new int [11];
void setup(){
  size( 400, 400 );
  textSize( 25 );
  for( int i=0; i<11; i++ ){
    scores[i] = 0;
  }
  String [] lines
    = loadStrings( "http://snakamura.org/teach/fms/scores.txt");
  for( int i=0; i<lines.length; i++ ){
    scores[ int( lines[i] ) ]++;
  }
}
```

演習提出：描画アプリ作成

明治大学総合数理学部
先端メディアサイエンス学科
中村研究室



- ボタンがが押されている間にマウスがあった場所に小さな円を描く．また，描いた情報をファイルに記録し，再現可能とする
 - なお，色々と機能拡張してもよい
- ファイル名： DrawApp



ヒント



```
import java.awt.Point;
ArrayList<Point> dots;

void setup(){
    size( 400, 400 );
    dots = new ArrayList<Point>();
    loadFromFile();
}

void draw(){
    background( 255 );
    if( mousePressed ){
        dots.add( new Point( mouseX, mouseY ) );
    }
    for(int i=0; i<dots.size(); i++){
        ellipse( dots.get(i).x, dots.get(i).y, 3, 3 );
    }
}

void mouseReleased() {
    saveToFile();
}
```

```
void saveToFile(){
    String [] lines = new String [dots.size()];
    // ここにファイル書き込みのコード
    saveStrings("data.csv", lines );
}

void loadFromFile(){
    String [] lines = loadStrings( "data.csv" );
    if( lines != null ){
        // ここにファイル読み込みのコード
    }
}
```



```
void saveToFile(){
    String [] lines = new String [dots.size()];
    for( int i=0; i<dots.size(); i++ ){
        Point dot = dots.get(i);
        lines[i] = dot.x + "," + dot.y;
    }
    saveStrings("data.csv", lines );
}

void loadFromFile(){
    String [] lines = loadStrings( "data.csv" );
    if( lines != null ){
        for( int i=0; i<lines.length; i++ ){
            String [] m = split( lines[i], ',' );
            Point d = new Point( int(m[0]), int(m[1]) );
            dots.add( d );
        }
    }
}
```

演習（提出: click_count）



ウィンドウ上でクリックした回数を表示するプログラムを作成せよ。なお，クリックした回数を保存し，次回起動した時にそのクリック回数から増やしていくようにせよ

- count.txt というファイルを作成
- click_count という変数を用意して，mousePressed() の度に値を増やす
- mousePressed() の度に，click_count の値をcount.txt に保存する
- 起動時（setup）の時に，count.txt の値をclick_count に読み込む